日本語プログラミング言語Mind８で十進数計算（符号付長整数計算）

## Mind独自の長整数を定義し四則演算するプログラム
入力データの型チェックや桁あふれなどの例外処理は、未実装です。
### 長整数型定義と部品
```
Ｌ整数型は　型紙
	Ｌ符号は　ワード変数
　	Ｌ数字は　変数
	Ｌ桁数は　変数
	全体は　Ｌ符号と　Ｌ桁数と　１００個の　Ｌ数字。

Ｌ数値変換とは
		Ｗ整数は　構造体　Ｌ整数型
　　　　Ｗ文字列は　文字列実体
	Ｗ文字列に　入れる
	Ｗ文字列の　左端文字で　事例をとり
　　　　　　'-'なら　　'-'を
　　　　　　'－'なら　　'-'を
			例外なら	'+'を
	事例終わり
			Ｗ整数の　Ｌ符号に　入れる
	Ｗ文字列の　文字数を　Ｗ整数の　Ｌ桁数に　入れる
	Ｗ文字列の　左端文字が　'-'と　等しい
　　または
	Ｗ文字列の　左端文字が　'－'と　等しい
	または
	Ｗ文字列の　左端文字が　'+'と　等しい
	または
	Ｗ文字列の　左端文字が　'＋'と　等しい
		ならば　Ｗ整数の　Ｌ桁数を　一つ減少し
		つぎに
	Ｗ整数の　Ｌ桁数　回数指定し
		Ｗ文字列から　一文字右側切り出し　
		一文字数値変換し　捨て　Ｗ整数の　Ｌ数字（回数）に　入れる
	繰り返す
	Ｗ整数を　返す。

Ｌ文字列変換とは
		Ｗ整数は　構造体情報
		Ｗ文字列は　文字列実体
		Ｗ桁数は　変数
		Ｗ桁は　変数
	Ｗ整数に　入れる		
	Ｗ文字列を　クリアする
	Ｗ整数の　Ｌ符号を　Ｗ文字列に　一文字追加し
	Ｗ整数の　Ｌ桁数を　Ｗ桁数に　入れる
	Ｗ桁数を　Ｗ桁に　入れる
	Ｗ桁数　回数指定し
		Ｗ整数の　Ｌ数字（Ｗ桁）を　文字列変換し
		Ｗ文字列に　追加する
		Ｗ桁を　一つ減少
	繰り返す
	Ｗ文字列を　返す。
 
整数をＬ整数に変換とは　（数値　→　Ｌ整数）
	文字列変換し　Ｌ数値変換し　返す。

Ｌゼロは　（Ｌ整数）
	　"0"を　Ｌ数値変換し　返す。
```
## 大小比較、部分数字列抽出など
```
Ｌ大小比較とは　（Ｌ整数、Ｌ整数　→　1/0/-1）
		Ｗ整数Ａは　構造体情報
　　　　Ｗ整数Ｂは　構造体情報
		Ｗ桁数Ａは　変数
		Ｗ桁数Ｂは　変数
　　　　Ｗ桁は　　　変数
	Ｗ整数Ａと　Ｗ整数Ｂに　入れる
	Ｗ整数Ａの　Ｌ桁数を　Ｗ桁数Ａに　入れる
	Ｗ整数Ｂの　Ｌ桁数を　Ｗ桁数Ｂに　入れる
	Ｗ桁数Ａが　Ｗ桁数Ｂより　大きい
		ならば　１を　返し　終わり
		つぎに
	Ｗ桁数Ｂが　Ｗ桁数Ａより　大きい
		ならば　-１を　返し　終わり
		つぎに
	Ｗ桁数Ａを　Ｗ桁に　入れる
	Ｗ桁数Ａ　回数指定し
		Ｗ整数Ａの　Ｌ数字（Ｗ桁）が　Ｗ整数Ｂの　Ｌ数字（Ｗ桁）より　大きい
			ならば　　１を　返し　終わり
			つぎに
		Ｗ整数Ａの　Ｌ数字（Ｗ桁）が　Ｗ整数Ｂの　Ｌ数字（Ｗ桁）より　小さい
			ならば　　-１を　返し　終わり
			つぎに
		Ｗ桁を　一つ減少する
	繰り返す
	０を　返す。
Ｌ指定桁数の上位部分整数とは　（L整数、数値　→　Ｌ整数）
		Ｗ桁数Ａは　変数
		Ｗ整数Ａは　構造体情報
		Ｗ整数Ｂは　構造体　Ｌ整数型
		Ｗ桁数Ｂは　変数
　　　　Ｗ桁Ａは　　変数
　　　　Ｗ桁Ｂは　　変数	
	Ｗ整数Ａと　Ｗ桁数Ｂに　入れる　
　　Ｗ整数Ａの　Ｌ桁数を　Ｗ桁数Ａに　入れる
　　Ｗ桁数Ｂが　Ｗ桁数Ａより　大きい
		ならば　Ｗ桁数Ａを　Ｗ桁数Ｂに　入れる
		つぎに
　　Ｗ桁数Ａを　Ｗ桁Ａに　入れる
	Ｗ桁数Ｂを　Ｗ桁Ｂに　入れる
	Ｗ桁数Ｂ　回数指定し
		Ｗ整数Ａの　Ｌ数字（Ｗ桁Ａ）を　
　　　　　　　　Ｗ整数Ｂの　Ｌ数字（Ｗ桁Ｂ）に　入れる
		Ｗ桁Ａを　一つ減少する
		Ｗ桁Ｂを　一つ減少する
	繰り返す
　　Ｗ桁数Ｂを　Ｗ整数Ｂの　Ｌ桁数に　入れる
	Ｗ整数Ｂを　返す。　　　　　　　
```
### 前ゼロ抑止,繰り上がり処理
```
Ｌ前ゼロ抑止とは　（Ｌ整数　→　Ｌ整数）
        Ｗ整数は　構造体　　Ｌ整数型
		Ｗ桁数は　変数
	Ｗ整数に　入れる
	Ｗ整数の　Ｌ桁数を Ｗ桁数に　入れる　	
	ここから		
　　　　Ｗ桁数が　２　以上
		かつ
　　　　Ｗ整数の　Ｌ数字（Ｗ桁数）が　０と　等しい
			ならば	Ｗ桁数を　一つ減少する
			さもなければ
				打ち切り
			つぎに
	繰り返す　　
	Ｗ桁数を　Ｗ整数の　Ｌ桁数に　入れる
	Ｗ整数を　返す。
　　　　　　　
Ｌ繰り上がり処理とは　（Ｌ整数　→　Ｌ整数）
        Ｗ整数は　構造体情報
		Ｗ桁数は　変数
		Ｗ桁は　　変数
		Ｗ商は　　変数
		Ｗ余は　　変数
　　　　Ｗ値は　　変数
	Ｗ整数に　入れる
	Ｗ整数の　Ｌ桁数を Ｗ桁数に　入れる
　	Ｗ桁数から　一つ引いて　回数指定し
		Ｗ整数の　Ｌ数字（回数）が　１０　以上
　　　　　　　　　　　ならば　			
			Ｗ整数の　Ｌ数字（回数）を　１０での　余りと商をとり
　　　　　　　　	Ｗ整数の Ｌ数字（回数）と　Ｗ商に　入れる
			回数に　一つ加え　Ｗ桁に　入れる
			Ｗ整数の Ｌ数字（Ｗ桁）を　Ｗ商だけ　増加する
　　　　　　　　　　　　もう一度
		      つぎに		
		Ｗ整数の Ｌ数字（回数）が　０より　小さい
			ならば
				Ｗ整数の Ｌ数字（回数）に -1を　掛け　一を引いて
　　　　　　　　　　　　　　　	１０での　余りと商をとり　Ｗ余りと　Ｗ商に　入れる
				Ｗ商に　一つ加え　Ｗ値に　入れる
				回数に　一つ加え　Ｗ桁に　入れる
				Ｗ整数の Ｌ数字（Ｗ桁）を　Ｗ値だけ　減少する
　　　　　　　　Ｗ値に　１０を　掛け　Ｗ値に　入れ
　　　　　　　　Ｗ整数の Ｌ数字（回数）を　Ｗ値だけ　増加し			
			つぎに
	繰り返す	　　　
	ここから
		Ｗ整数の Ｌ数字（Ｗ桁数）が　１０　以上
			ならば
				Ｗ整数の Ｌ数字（Ｗ桁数）を　１０での　余りと商をとり　
				Ｗ整数の Ｌ数字（Ｗ桁数）と　Ｗ商に　入れる
　　　　　　　　　		Ｗ桁数を　一つ増加し
				Ｗ商を　Ｗ整数の Ｌ数字（Ｗ桁数）に　入れる
　　　　　　さもなければ
　				打ち切り
			つぎに
	繰り返す
	ここから		
　　　　Ｗ桁数が　２　以上
		かつ
　　　　Ｗ整数の　Ｌ数字（Ｗ桁数）が　０と　等しい
			ならば	Ｗ桁数を　一つ減少する
			さもなければ
				打ち切り
			つぎに
	繰り返す
	Ｗ桁数を　Ｗ整数の　Ｌ桁数に　入れる
	Ｗ整数を　返す。　　　　　　　
```
### 正数の加算、減算
```
Ｌ正数の加算結果とは　（Ｌ整数、Ｌ整数　→　Ｌ整数）
　　　　Ｗ整数Ａは　構造体情報
　　　　Ｗ整数Ｂは　構造体情報
		Ｗ整数Ｃは　構造体　Ｌ整数型
		Ｗ桁数Ａは　変数
		Ｗ桁数Ｂは　変数
		Ｗ和は　変数
		Ｗ桁数Ｃは　変数
	Ｗ整数Ａと　Ｗ整数Ｂに　入れる
	Ｗ整数Ａの　Ｌ桁数を　Ｗ桁数Ａに　入れる
	Ｗ整数Ｂの　Ｌ桁数を　Ｗ桁数Ｂに　入れる	
	Ｗ桁数Ａが　Ｗ桁数Ｂ　以上
		ならば　Ｗ桁数Ａを
		さもなければ
　　　　　　　　　　　　Ｗ桁数Ｂを
		つぎに
　　			Ｗ桁数Ｃに　入れる
	Ｗ桁数Ｃを　Ｗ整数Ｃの　Ｌ桁数に　入れる
	Ｗ桁数Ｃ　回数指定し
		Ｗ和を　クリアする
　　　　Ｗ桁数Ａが　回数　以上
			ならば　Ｗ整数Ａの　Ｌ数字（回数）だけ　Ｗ和を　増加し
			つぎに
		Ｗ桁数Ｂが　回数　以上
			ならば　Ｗ整数Ｂの　Ｌ数字（回数）だけ　Ｗ和を　増加し
			つぎに
		Ｗ和を　Ｗ整数Ｃの　Ｌ数字（回数）に　入れる
	繰り返す
	'+'を　Ｗ整数Ｃの　Ｌ符号に　入れる
　　Ｗ整数Ｃを　Ｌ繰り上がり処理し　返す。

Ｌ正数の減算結果とは　　（Ｌ整数、Ｌ整数　→　Ｌ整数）　
		Ｗ整数Ａは　構造体情報
　　	Ｗ整数Ｂは　構造体情報
		Ｗ整数Ｃは　構造体　Ｌ整数型
		Ｗ桁数Ａは　変数
		Ｗ桁数Ｂは　変数
		Ｗ和は　変数
		Ｗ桁数Ｃは　変数
	Ｗ整数Ａと　Ｗ整数Ｂに　入れる
	Ｗ整数Ａの　Ｌ桁数を　Ｗ桁数Ａに　入れる
	Ｗ整数Ｂの　Ｌ桁数を　Ｗ桁数Ｂに　入れる	
	Ｗ桁数Ａが　Ｗ桁数Ｂ　以上
		ならば　Ｗ桁数Ａを
		さもなければ
　　　　　　　　Ｗ桁数Ｂを
		つぎに
　　			Ｗ桁数Ｃに　入れる
	Ｗ桁数Ｃを　Ｗ整数Ｃの　Ｌ桁数に　入れる
	Ｗ桁数Ｃ　回数指定し
		Ｗ和を　クリアする
　　　　Ｗ桁数Ａが　回数　以上
			ならば　Ｗ整数Ａの　Ｌ数字（回数）だけ　Ｗ和を　増加し
			つぎに
		Ｗ桁数Ｂが　回数　以上
			ならば　Ｗ整数Ｂの　Ｌ数字（回数）だけ　Ｗ和を　減少し
			つぎに
		Ｗ和を　Ｗ整数Ｃの　Ｌ数字（回数）に　入れる
	繰り返す
　	Ｗ整数Ｃを　Ｌ繰り上がり処理し　返す。
```
### 加算、減算
```
Ｌ加算結果とは　（Ｌ整数、Ｌ整数　→　Ｌ整数）
　　　　Ｗ整数Ａは　構造体情報
　　　　Ｗ整数Ｂは　構造体情報
		Ｗ整数Ｃは　構造体　Ｌ整数型
		Ｗ整数Ａの絶対値は　構造体　Ｌ整数型
　　　　Ｗ整数Ｂの絶対値は　構造体　Ｌ整数型
		Ｗ大小は　変数
	Ｗ整数Ａと　Ｗ整数Ｂに　入れる
	Ｗ整数Ａを　Ｗ整数Ａの絶対値に　入れる
　　Ｗ整数Ａの絶対値の　Ｌ符号に　'+'を　入れる
	Ｗ整数Ｂを　Ｗ整数Ｂの絶対値に　入れる
　　Ｗ整数Ｂの絶対値の　Ｌ符号に　'+'を　入れる
	Ｗ整数Ａの　Ｌ符号と　Ｗ整数Ｂの　Ｌ符号が　等しい
		ならば　Ｗ整数Ａの絶対値と　Ｗ整数Ｂの絶対値の　Ｌ正数の加算結果を
				Ｗ整数Ｃに　入れる
				Ｗ整数Ａの　Ｌ符号を　Ｗ整数Ｃの　Ｌ符号に　入れる
		さもなければ
				Ｗ整数Ａの絶対値と　Ｗ整数Ｂの絶対値の　Ｌ大小比較し
				Ｗ大小に　入れる
				Ｗ大小が　０と　等しい
					ならば　Ｌゼロを　返し　終わり
					つぎに
				Ｗ大小が　１と　等しい
					ならば　Ｗ整数Ａの絶対値から　Ｗ整数Ｂの絶対値の
						　	Ｌ正数の減算結果を　Ｗ整数Ｃに　入れる		　
							Ｗ整数Ａの　Ｌ符号を　Ｗ整数Ｃの　Ｌ符号に　入れる

					さもなければ
							Ｗ整数Ｂの絶対値から　Ｗ整数Ａの絶対値の
						　	Ｌ正数の減算結果を　Ｗ整数Ｃに　入れる		　
							Ｗ整数Ｂの　Ｌ符号を　Ｗ整数Ｃの　Ｌ符号に　入れる
					つぎに
		つぎに
　　Ｗ整数Ｃを　返す。

Ｌ減算結果とは　　（Ｌ整数、Ｌ整数　→　Ｌ整数）　※答えが負の場合は未対応
　　　　Ｗ整数Ａは　構造体情報
　　　　Ｗ整数Ｂは　構造体情報
		Ｗ整数Ｃは　構造体　Ｌ整数型
		Ｗ整数Ａの絶対値は　構造体　Ｌ整数型
　　　　Ｗ整数Ｂの絶対値は　構造体　Ｌ整数型
		Ｗ大小は　変数
	Ｗ整数Ａと　Ｗ整数Ｂに　入れる
	Ｗ整数Ａを　Ｗ整数Ａの絶対値に　入れる
　　Ｗ整数Ａの絶対値の　Ｌ符号に　'+'を　入れる
	Ｗ整数Ｂを　Ｗ整数Ｂの絶対値に　入れる
　　Ｗ整数Ｂの絶対値の　Ｌ符号に　'+'を　入れる
	Ｗ整数Ａの　Ｌ符号と　Ｗ整数Ｂの　Ｌ符号が　異なる
		ならば　Ｗ整数Ａの絶対値と　Ｗ整数Ｂの絶対値の　Ｌ正数の加算結果を
				Ｗ整数Ｃに　入れる
				Ｗ整数Ａの　Ｌ符号を　Ｗ整数Ｃの　Ｌ符号に　入れる
		さもなければ
				Ｗ整数Ａの絶対値と　Ｗ整数Ｂの絶対値の　Ｌ大小比較し
				Ｗ大小に　入れる
				Ｗ大小が　０と　等しい
					ならば　Ｌゼロを　返し　終わり
					つぎに
				Ｗ大小が　１と　等しい
					ならば　Ｗ整数Ａの絶対値から　Ｗ整数Ｂの絶対値の
						　	Ｌ正数の減算結果を　Ｗ整数Ｃに　入れる		　
							Ｗ整数Ａの　Ｌ符号を　Ｗ整数Ｃの　Ｌ符号に　入れる

					さもなければ
							Ｗ整数Ｂの絶対値から　Ｗ整数Ａの絶対値の
						　	Ｌ正数の減算結果を　Ｗ整数Ｃに　入れる
							Ｗ整数Ａの　Ｌ符号が　'+'と　等しい
								ならば　'-'を
								さもなければ
										'+'を
								つぎに		　
												Ｗ整数Ｃの　Ｌ符号に　入れる

					つぎに
		つぎに
　　Ｗ整数Ｃを　返す。
```
### 掛け算
```
Ｌ掛け算結果とは　（Ｌ整数、Ｌ整数　→　Ｌ整数）
　　　　Ｗ整数Ａは　構造体情報
　　　　Ｗ整数Ｂは　構造体情報
		Ｗ整数Ｃは　構造体　Ｌ整数型
		Ｗ桁数Ａは　変数
		Ｗ桁数Ｂは　変数
　　　　Ｗ桁は　　　変数
	Ｗ整数Ａと　Ｗ整数Ｂに　入れる
	Ｗ整数Ａの　Ｌ桁数を　Ｗ桁数Ａに　入れる
	Ｗ整数Ｂの　Ｌ桁数を　Ｗ桁数Ｂに　入れる
	Ｗ整数Ａの　Ｌ符号と　Ｗ整数Ｂの　Ｌ符号が　等しい
		ならば　'+'を
		さもなければ　'-'を
		つぎに
				Ｗ整数Ｃの　Ｌ符号に　入れる	
	Ｗ桁数Ａに　Ｗ桁数Ｂを　加え　一を引いて　Ｗ整数Ｃの　Ｌ桁数に　入れる
	Ｗ整数Ｃの　Ｌ桁数　回数指定し
		Ｗ整数Ｃの　Ｌ数字（回数）を　クリアする
	繰り返す
	Ｗ桁数Ａ　回数指定し
		Ｗ桁数Ｂ　回数指定し
			外側回数と　回数を　加え 一つ引いて　Ｗ桁に　入れる
　　　　　　Ｗ整数Ａの　Ｌ数字（外側回数）と
　　　　　　Ｗ整数Ｂの　Ｌ数字（回数）を　掛け　
　　　　　　Ｗ整数Ｃの　Ｌ数字（Ｗ桁）を　増加する
		繰り返す
	繰り返す	
　　Ｗ整数Ｃを　Ｌ繰り上がり処理し　返す。

Ｌ整数と整数の掛け算結果とは　（Ｌ整数、整数　→　Ｌ整数）
　　　　Ｗ整数Ａは　構造体情報
　　　　Ｗ整数Ｂは　変数
		Ｗ整数Ｃは　構造体　Ｌ整数型
		Ｗ桁数Ａは　変数
	Ｗ整数Ａと　Ｗ整数Ｂに　入れる
	Ｗ整数Ａの　Ｌ桁数を　Ｗ桁数Ａに　入れる
	Ｗ桁数Ａを　Ｗ整数Ｃの　Ｌ桁数に　入れる
	Ｗ桁数Ａ　回数指定し
　　　　Ｗ整数Ａの　Ｌ数字（回数）と　Ｗ整数Ｂを　掛け　
　　　　Ｗ整数Ｃの　Ｌ数字（回数）に　入れる
	繰り返す
	　'+'を　Ｗ整数Ｃの　Ｌ符号に　入れる
　	Ｗ整数Ｃを　Ｌ繰り上がり処理し　返す。
```
### 割り算
```
一桁目に挿入とは　（整数、Ｌ整数　→　Ｌ整数）
		Ｗ余りは　構造体情報
　　　　Ｗ挿入数は　変数
　　	Ｎ桁は　変数
		Ｎ次の桁は　変数
	Ｗ挿入数と　Ｗ余りに　　入れる　　　　　　　　　　
	Ｗ余りの　Ｌ桁数を　Ｎ桁に　入れる
	Ｎ桁が　１と　等しく
	かつ
　　Ｗ余りの　Ｌ数字（１）が　０と　等しい
		ならば　Ｗ挿入数を　Ｗ余りの　Ｌ数字（１）に　入れる
　　　　　　　　終わり
		つぎに　　　　　　　　　　　　　　
	Ｗ余りの　Ｌ桁数を　回数指定し
		Ｎ桁に　一つ加え　Ｎ次の桁に　入れる
　　　　Ｗ余りの　Ｌ数字（Ｎ桁）を　
		Ｗ余りの　Ｌ数字（Ｎ次の桁）に　入れる
　		Ｎ桁を　一つ減少する
　	繰り返す		
	Ｗ挿入数を　Ｗ余りの　Ｌ数字（１）に　入れる
	Ｗ余りの　Ｌ桁数を　一つ増加する。

Ｌ正数の余りと商とは　（Ｌ整数、Ｌ整数　→　Ｌ整数、Ｌ整数）
		Ｗ整数Ａは　構造体情報
		Ｗ整数Ｂは　構造体情報
		Ｗ整数Ｃは　構造体　Ｌ整数型
		Ｗ整数Ｘは　構造体　Ｌ整数型
		Ｗ余は　　構造体　Ｌ整数型
		Ｗ商は　構造体　Ｌ整数型
		ＮＡは　変数
		ＮＢは　変数
		Ｄは　　変数
　　　　桁数Ｅは　　変数
        ｉは　変数
		Ｗ桁は　　変数
	Ｗ整数Ａと　Ｗ整数Ｂに　入れる
	Ｗ整数Ａの　Ｌ桁数を　ＮＡに　入れる
	Ｗ整数Ｂの　Ｌ桁数を　ＮＢに　入れる　
	ＮＡが　ＮＢより　小さい
		ならば
			Ｗ整数Ａと　Ｌゼロを　返し　終わり		
		つぎに   
	Ｗ整数Ａの　ＮＢだけ　Ｌ指定桁数の上位部分整数を　Ｗ整数Ｃに　入れる 
　　Ｗ整数Ｃと　Ｗ整数Ｂを　Ｌ大小比較し 　０　以上
		ならば　[Ｄ　:＝ ＮＡ　-　ＮＢ　+　1　]
				[桁数Ｅ　:＝ ＮＢ]
		さもなければ
				[Ｄ　:＝ ＮＡ　-　ＮＢ　]
				[桁数Ｅ　:＝ ＮＢ　+　１]
		つぎに
	Ｄが　０と　等しい
		ならば　Ｗ整数Ａと　Ｌゼロを　返し　終わり
		つぎに
　　Ｗ整数Ａの　桁数Ｅだけ　　Ｌ指定桁数の上位部分整数を　Ｗ余りに　入れる
    Ｗ余りの　Ｌ符号に　'+'を　入れる
	Ｗ商を　クリアする
	Ｄを　Ｗ商の　Ｌ桁数に　入れる
	 Ｗ商の　Ｌ符号に　'+'を　入れる
	Ｄ　回数指定し
		[ｉ := Ｄ -　回数 + １]
		Ｗ商の　Ｌ数字（ｉ）に　９を　入れる
		９　回数指定し
　　　　　　Ｗ整数Ｂと　回数の　Ｌ整数と整数の掛け算結果を　Ｗ整数Ｘに　入れる	
　　　　	Ｗ整数Ｘと　Ｗ余りを　Ｌ大小比較し　１と　等しい
				ならば　回数から　一つ引いて　Ｗ商の　Ｌ数字（ｉ）に　入れる		
						打ち切り
				つぎに
		繰り返す
		Ｗ整数Ｂと　Ｗ商の　Ｌ数字（ｉ）の　Ｌ整数と整数の掛け算結果を
　　　　Ｗ整数Ｘに　入れる
	　	Ｗ余りから　Ｗ整数Ｘを　Ｌ正数の減算結果を Ｗ余りに　入れる
		ｉが　２　以上
			ならば　
				ｉから　一つ引いて　Ｗ桁に　入れる
            	Ｗ整数Ａの　Ｌ数字（Ｗ桁）を　Ｗ余りに　一桁目に挿入し
			つぎに
	繰り返す
	Ｗ余と　Ｗ商を　返す。

Ｌ余りと商とは　（Ｌ整数、Ｌ整数　→　Ｌ整数、Ｌ整数）
		Ｗ整数Ａは　構造体情報
		Ｗ整数Ｂは　構造体情報
		Ｗ整数Ａの絶対値は　構造体　Ｌ整数型
　　　　Ｗ整数Ｂの絶対値は　構造体　Ｌ整数型
		符号Ａは　ワード変数
		符号Ｂは　ワード変数
		Ｗ余は　構造体　Ｌ整数型
		Ｗ商は　構造体　Ｌ整数型
	Ｗ整数Ａと　Ｗ整数Ｂに　入れる
	Ｗ整数Ａを　Ｗ整数Ａの絶対値に　入れる
　　Ｗ整数Ａの絶対値の　Ｌ符号に　'+'を　入れる
	Ｗ整数Ｂを　Ｗ整数Ｂの絶対値に　入れる
　　Ｗ整数Ｂの絶対値の　Ｌ符号に　'+'を　入れる 
	Ｗ整数Ａの絶対値を　Ｗ整数Ｂの絶対値の　
　　	Ｌ正数の余りと商を　Ｗ余りと　Ｗ商に　入れる  
	Ｗ整数Ａの　Ｌ符号が　'+'と　等しい
		ならば
			Ｗ整数Ｂの　Ｌ符号が　'+'と　等しい
				ならば　'+'を　Ｗ余りの　Ｌ符号に　入れる
　　　　　　　　　　　　'+'を　Ｗ商の　Ｌ符号に　入れる
				さもなければ
　　　　　　　　　　　　'+'を　Ｗ余りの　Ｌ符号に　入れる
　　　　　　　　　　　　'-'を　Ｗ商の　Ｌ符号に　入れる
				つぎに
		さもなければ
			Ｗ整数Ｂの　Ｌ符号が　'+'と　等しい
				ならば　'-'を　Ｗ余りの　Ｌ符号に　入れる
　　　　　　　　　　　　'-'を　Ｗ商の　Ｌ符号に　入れる
				さもなければ
　　　　　　　　　　　　'-'を　Ｗ余りの　Ｌ符号に　入れる
　　　　　　　　　　　　'+'を　Ｗ商の　Ｌ符号に　入れる
				つぎに
		つぎに
	Ｗ余と　Ｗ商を　返す。　　
```
### テスト
```
Ｌ整数１は　構造体　Ｌ整数型。
Ｌ整数２は　構造体　Ｌ整数型。
Ｌ整数３は　構造体　Ｌ整数型。
Ｌ整数４は　構造体　Ｌ整数型。
Ｌ整数５は　構造体　Ｌ整数型。
Ｌ整数６は　構造体　Ｌ整数型。
Ｌ整数７は　構造体　Ｌ整数型。

メインは　　　　　　　　（ ・ －－＞ ・ ）
　　Ｗ文字列は　文字列実体
	「２つの自然数の四則演算をします。（空入力で終了）」と 表示し　改行し
	ここから
		「自然数１＝」と　表示し
　　　　　　　　　文字列入力し　Ｗ文字列に　入れる
		Ｗ文字列が　空列？
			ならば　打ち切り
			つぎに
 		Ｗ文字列を　Ｌ数値変換し　Ｌ整数１に　入れる
		「自然数２＝」と　表示し
			文字列入力し　Ｗ文字列に　入れる
		Ｗ文字列を　Ｌ数値変換し　Ｌ整数２に　入れる
			Ｌ整数１と　Ｌ整数２の　Ｌ加算結果を　Ｌ整数３に　入れる
		「　　　和＝」と　表示し　Ｌ整数３を　Ｌ文字列変換し　表示し　改行する
			Ｌ整数１と　Ｌ整数２の　Ｌ減算結果を　Ｌ整数４に　入れる
		「　　　差＝」と　表示し　Ｌ整数４を　Ｌ文字列変換し　表示し　改行する
			Ｌ整数１と　Ｌ整数２の　Ｌ掛け算結果を　Ｌ整数５に　入れる
		「　　　積＝」と　表示し　Ｌ整数５を　Ｌ文字列変換し　表示し 改行し
			Ｌ整数１と　Ｌ整数２の　Ｌ余りと商を　Ｌ整数６と　Ｌ整数７に　入れる	 		「　　　商＝」と　表示し　Ｌ整数７を　Ｌ文字列変換し　表示し
		「　余＝」と　表示し　Ｌ整数６を　Ｌ文字列変換し　表示し
		　改行する		
	繰り返す。
```
実行結果
```
日本語プログラミング言語 Ｍｉｎｄ Version 8.10 for Windows
          Copyright(C) 1985 Scripts Lab. Inc.
コンパイル中 .. 終了
Coping.. c:\mind9\pmind\bin\mindex.exe --> lnumh2.exe

c:\mind8\pmind\sample>lnumh2
２つの自然数の四則演算をします。（空入力で終了）
自然数１＝11111111111111
自然数２＝11111111
　　　和＝+11111122222222
　　　差＝+11111100000000
　　　積＝+123456788888887654321
　　　商＝+1000000　余＝+111111
自然数１＝123456789
自然数２＝123456789
　　　和＝+246913578
　　　差＝+0
　　　積＝+15241578750190521
　　　商＝+1　余＝+0
自然数１＝123456789
自然数２＝234567
　　　和＝+123691356
　　　差＝+123222222
　　　積＝+28958888625363
　　　商＝+526　余＝+74547
自然数１＝234567
自然数２＝526
　　　和＝+235093
　　　差＝+234041
　　　積＝+123382242
　　　商＝+445　余＝+497
自然数１＝123382242
自然数２＝74547
　　　和＝+123456789
　　　差＝+123307695
　　　積＝+9197775994374
　　　商＝+1655　余＝+6957
自然数１＝3
自然数２＝-2
　　　和＝+1
　　　差＝+5
　　　積＝-6
　　　商＝-1　余＝+1
自然数１＝-3
自然数２＝2
　　　和＝-1
　　　差＝-5
　　　積＝-6
　　　商＝-1　余＝-1
自然数１＝-3
自然数２＝-2
　　　和＝-5
　　　差＝-1
　　　積＝+6
　　　商＝+1　余＝-1
   ```
   ### 割り算の余りについて
   mindの処理単語「余りと商」に合わせて、負数の余りを使っています。